


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

L.1000

PER IL TUO **Commodore**

Una pubblicazione della J.soft editrice

sped. in abb. post. Gruppo Il/70



PAPER

soft

28

Anno 2 - N° 28 - 19 luglio 1985

VIC-20

**La corona
di pietra**

VIC-20

Guerra spaziale

**Titolatore di
cassette**

C16

Explorer

C16

C64

**Decidere
L'accoppiata**

C64

STM

Guida per l'input dei programmi versioni VIC 20 e C64

Notate che i listati contengono 'parole' racchiuse tra parentesi graffe { }. Tali parole rappresentano caratteri di controllo come mostrato nel sottostante riquadro. Se sono precedute da un numero, questo indica il numero di volte che quel tasto deve essere premuto. Se il simbolo è sottolineato deve essere premuto contemporaneamente a SHIFT mentre se è racchiuso da {} deve essere premuto con-

temporaneamente al tasto COMMODORE. Inoltre, se tra parentesi graffa si trova un carattere alfabetico "solitario" questo dovrà essere premuto contemporaneamente al tasto CTRL.

Con questo sistema di codifica, sarà molto più agevole copiare i listati senza faticose e dubbie interpretazioni di caratteri grafici e di controllo del cursore o dei colori.

{CLR}	SHIFT CLR/HOME		{CYN}	CTRL 4		{<7>}	Q 7	
{HOME}	CLR/HOME		{PUR}	CTRL 5		{<8>}	Q 8	
{SU}	SHIFT {} CSR {}		{GRN}	CTRL 6		{F1}	F 1	
{GIU}	{} CSR {}		{BLU}	CTRL 7		{F2}	F 2	
{SIN}	SHIFT {} CSR =>		{YEL}	CTRL 8		{F3}	F 3	
{DES}	{} CSR =>		{<1>}	Q 1		{F4}	F 4	
{RVS}	CTRL 4		{<2>}	Q 2		{F5}	F 5	
{OFF}	CTRL 5		{<3>}	Q 3		{F6}	F 6	
{BLK}	CTRL 1		{<4>}	Q 4		{F7}	F 7	
{WHT}	CTRL 2		{<5>}	Q 5		{F8}	F 8	
{RED}	CTRL 3		{<6>}	Q 6				

Per ricevere gli arretrati di **PAPER** compilare il seguente tagliando:



J.soft **Compilare ed inviare in busta chiusa a:**
viale Restelli, 5 - 20124 Milano - Tel. 02/6888228-683797-6880841-2-3

Nome _____

Cognome _____

Indirizzo _____

CAP _____ Città _____ Prov. _____

Desidero ricevere i seguenti numeri arretrati di

PAPER _____ Anno _____

PAPER dedicato a _____ Anno _____

a L. 2.000 cad. per un totale di L. _____ a mezzo

assegno allegato

contanti allegati





VIC-20

- 7** **La Corona di Pietra (16K RAM)**
prima parte
di E. Ceppi
- 13** **Guerra spaziale (3K RAM e joystick)**
di W. Hamker trad. e adatt. di F. Sarcina

C64

- 22** **Decidere**
di C. Antoniutti
- 26** **L'accoppiata**
di F. Giurco trad. e adatt. di S. Albarelli
- 29** **STM**
di P. Gavelor trad. e adatt. di P. Maffei

C16-plus/4

- 18** **Titolatore di cassette (stampante)**
di A. Van Aardt trad. e adatt. di F. Sarcina
- 20** **Explorer**
di L. Dolphin trad. e adatt. di S. Dolphin

J.soft s.r.l.

**DIREZIONE, REDAZIONI
E AMMINISTRAZIONE**

Via Restelli, 5
20124 Milano
Tel.: 68.88.228-68.37.97

DIRETTORE RESPONSABILE:

Pietro Dell'Orco

COORDINAMENTO TECNICO:

Mauro Cristulib Grizzi

REDAZIONE:

Franco Sarcina

GRAFICA E IMPAGINAZIONE

Margherita La Noce
Ivana Rossi
Raffaella Toffolatti

FOTOCOPOSIZIONE:

Graphotek
Via Astesani, 16 - Milano
Tel. 64.80.397

CONTABILITÀ:

Giulia Pedrazzini
Flavia Bonaiti

**AUTORIZZAZIONE ALLA
PUBBLICITÀ:**

Tribunale di Milano n° 178
del 14.04.1984

STAMPA:

Intergrafica - Pioltello (MI)
Rivista associata
all'Unione stampa
Periodica Italiana



PUBBLICITÀ

Concessionario per l'Italia e l'Estero
J-Advertising s.r.l.

V.le Restelli, 5
20124 MILANO
Tel. (02)

68.82.895-68.80.606-68.87.233
Tlx 316213 REINA I
Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero:
SODIP - Via Zuretti, 25
20125 MILANO

Spedizione in abbonamento
postale Gruppo II/70
Prezzo della rivista L. 1.000
Numero arretrati L. 2.000

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI



Stop Bug

Se non l'hai già fatto, memorizza e salva su nastro o disco la versione di Stop Bug per il tuo computer. Questo programma ti permette, una volta caricato ed attivato con Run, di digitare i listati di Papersoft Commodore senza introdurre errori di battitura.

Come avrai notato, le linee dei nostri programmi terminano con la parola ".rem" seguita da un numero, detto checksum; non copiare questa parte della linea, che serve solo per tua informazione. Quando Stop Bug è in funzione, infatti, tutte le volte che premi il tasto Return dopo aver digitato una linea di programma, appare nell'angolo superiore sinistro dello schermo un numero, che deve coincidere con il checksum da noi pubblicato (sempre che tu non abbia usato

comandi abbreviati, come ? per Print).

Usando Stop Bug tieni conto del fatto che gli spazi non influiscono sui checksum, per cui fai particolare attenzione alle istruzioni Print dove sono spesso presenti. Inoltre eventuali inversioni di caratteri non vengono segnalate.

Un altro metodo per utilizzare Stop Bug, soprattutto se sei abituato ad usare le abbreviazioni dei comandi Basic, è controllare i checksum listando il programma e battendo Return su ogni linea.

Se intendi sospendere il lavoro di digitazione, puoi salvare su nastro o disco il programma che stai memorizzando senza che Stop Bug interferisca; al momento di riprendere il lavoro ricordati però di caricare ed attivare Stop Bug prima di ricaricare il tuo programma.

Per disabilitare Stop Bug premi Run/Stop e Restore (su C64 e VIC 20) oppure Run/Stop e Reset (su C16 e Plus/4). Per riattivarlo batti:

*SYS PEEK(56)*256 per il VIC 20;
SYS 16128 per il C16 e il PLUS/4;
SYS 40704 per il C 64.*

STOP BUG per C64 e VIC 20

```
100 POKE55,0:POKE56,PEEK(56)-1:R=PEEK(56)*256:PRINT"{CLR}
ATTENDERE PREGO..."
110 FORI=RTOR+132:READA:CK=CK+A:IFA=999THENA=R/256
120 POKEI,A:NEXT
130 IFCK<>22689THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
TA":END
140 SYS(R):PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIVAT
O.":NEW
200 DATA173,36,3,201,32,208,1,96,141
210 DATA33,999,173,37,3,141,34,999,169
220 DATA32,141,36,3,169,999,141,37,3
230 DATA169,0,133,254,96,32,87,241,133
240 DATA251,134,252,132,253,8,201,13,240
250 DATA17,201,32,240,5,24,101,254,133
260 DATA254,165,251,166,252,164,253,40,96
270 DATA169,13,32,210,255,165,214,141,251
280 DATA3,206,251,3,169,0,133,216,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,254,169,0
310 DATA133,254,172,33,999,192,87,208,6
320 DATA32,205,189,76,117,999,32,205,221
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,214,76,55,999
```

STOP BUG per C16 e PLUS/4

```
100 POKE55,0:POKE56,63:PRINT"{CLR}ATTENDERE PREGO..."
110 FORI=16128TO16260:READA:CK=CK+A:POKEI,A:NEXT
120 IFCK<>14248THENPRINT"{GIU'}ERRORE NELLE ISTRUZIONI DA
TA":END
130 SYS16128:PRINT"{CLR}{ 2 GIU'}{RVS}STOP BUG{OFF} ATTIV
ATO.":NEW
200 DATA173,34,3,201,32,208,1,96,141
210 DATA33,63,173,35,3,141,34,63,169
220 DATA32,141,34,3,169,63,141,35,3
230 DATA169,0,133,6,96,32,232,235,133
240 DATA3,134,4,132,5,8,201,13,240
250 DATA17,201,32,240,5,24,101,6,133
260 DATA6,165,3,166,4,164,5,40,96
270 DATA169,13,32,210,255,165,205,141,251
280 DATA3,206,251,3,169,0,133,207,169
290 DATA19,32,210,255,169,18,32,210,255
300 DATA169,58,32,210,255,166,6,169,0
310 DATA133,6,172,33,63,192,232,208,6
320 DATA32,95,164,76,117,63,32,7,208
330 DATA169,32,32,210,255,32,210,255,173
340 DATA251,3,133,205,76,55,63
```

IN EDICOLA

SUPER

sinc

SUPER

C-COMMODORE

N°7



**SuperSinc e
Super Commodore
sono idee**

J.soft EDITRICE

**CON
CASSETTA**

La Corona di Pietra

Prima parte

Siamo finalmente in grado di proporre ai nostri lettori un appassionante adventure game per il VIC 20. La mole del programma, presuppone l'uso di una espansione da 16K, ed infatti il listato è talmente lungo che siamo stati costretti a dividerlo in due parti: la prima appare su questo numero, mentre la seconda verrà pubblicata la settimana prossima.

In questa avventura voi impersonate un Principe, al quale l'infido consigliere ha sottratto la Corona di Pietra, simbolo del potere, per poi diventare un feroce tiranno. Per impedirvi di riassumere il comando egli ha nascosto la Corona in un luogo sconosciuto e pieno di pericoli; ma voi coraggiosamente sfidate il pericolo e iniziate la ricerca, per poter risalire sul trono e liberare il vostro popolo dall'oppressione del tiranno.

Per chi non avesse troppa dimestichezza con i giochi d'avventura, spieghiamo che per risolverli occorre principalmente molta pazienza: qui non contano i riflessi, ma bensì la riflessione e l'attenzione anche al più piccolo dei particolari. Dovete immaginare di essere il cervello, mentre il computer è il vostro corpo; esso vi descrive dove vi trovate, cosa c'è intorno a voi e la vostra situazione, e voi, il cervello, dovete dirgli cosa fare. Il computer, allora, cercherà di obbedirvi nei limiti del possibile.

Un'azione può essere eseguibile in alcuni casi ed in altri no, può richiedere di avere con sé uno o più oggetti, e così via. Ricordate comunque che la soluzione esiste, anche se a volte non è di immediata comprensione.

Il vocabolario di cui dispone il computer comprende una sessantina tra nomi e verbi, ed i luoghi a cui è possibile accedere sono ben 32. I messaggi possono essere costituiti da parole singole (comandi) o da frasi del tipo "verbo + articolo/preposizione + oggetto". I comandi sono:

nord (abbreviabile con N), sud (S), est (E), ovest (O), scendi, inventario (I), aiuto!, mi arrendo, guarda/osserva ().*

I verbi nelle frasi possono essere: abbassa, abbatti, accendi, afferra, alimenta, apri, butta, esamina, getta, gira, guarda, nutri, osserva, piglia, posa, prendi, rompi, ruota, sali, scendi, spacca, spegni, spezza, sposta, taglia, uccidi, vai.

I termini presenti in entrambi gli elenchi sono parole che possono assumere significati diversi a seconda del contesto, purché siano impiegate come impone la grammatica italiana.

Per ora il vostro compito è solo quello di digitare e salvare su nastro o disco la prima parte del programma; la settimana prossima potrete completare l'opera e dare il faticoso RUN.

```

10 REM *****                               :rem 145
20 REM * LA CORONA *                           :rem 233
30 REM * DI PIETRA *                           :rem 237
40 REM * (C) 1984 *                             :rem 6
50 REM *ENRICO CEPPI*                           :rem 206
60 REM *****                               :rem 150
70 DIMNO$(27),SE$(27),AR$(27):GOSUB3120        :rem 124
80 SE%=1                                         :rem 147
90 PO$="":P1$="":P2$="":P3$="":N%=-1:S%=-1:E%=-1:O%=-1:GI
%=-1                                           :rem 88
100 ONSE%+1GOSUB2100,2130,2170,2200,2240,2290,2330,2360,2
380,2400,2420,2450                             :rem 16
110 ON-(SE%-11)*(SE%>11)GOSUB2480,2500,2530,2560,2590,261
0,2630,2660,2690,2720                          :rem 215
120 ON-(SE%-21)*(SE%>21)GOSUB2770,2790,2830,2850,2880,292
0,2940,2980,3020                                :rem 248
130 IFSE%=3AND(SE%(27)=3ORSE%(27)=4)THENE%=4   :rem 238
140 IFSE%=4AND(SE%(27)=3ORSE%(27)=4)THENO%=3   :rem 249
150 IF(SE%=14ORSE%=15ORSE%=17ORSE%=18ORSE%=19ORSE%=20)AND
(TO%<1ORSE%(5)<>-1)THEN3290                     :rem 210
160 PRINT"{CLR}";P0$;P1$;P2$;P3$               :rem 125
170 IF(SE%=3ORSE%=4)AND(SE%(27)=3ORSE%(27)=4)THENPRINT"PO
SSO PASSARE SUL{ 5 SPAZI}TRONCO!!             :rem 148
180 PRINT"{RVS}POSSO ANDARE A:{OFF}           :rem 166
190 IFN%<>-1THENPRINT"NORD ";                   :rem 226
200 IFS%<>-1THENPRINT"SUD ";                   :rem 152
210 IFE%<>-1THENPRINT"EST ";                   :rem 139
220 IFO%<>-1THENPRINT"OVEST ";                 :rem 59
230 IF(SE%=25ANDSE%(16)=-1)OR(SE%(3)=SE%ANDO%<>-1)THENPRI
NT"SU ";                                         :rem 232
240 IFGI%<>-1THENPRINT"GIU'";                  :rem 249
250 PRINT:CO%=0:FORI=0TO27:                     :rem 159
260 IFSE%(I)<>SE%THEN310                         :rem 120
270 IFCO%=0THENCO%=1:PRINT"{RVS}VEDO:         :rem 3
280 IFSE%(1)=SE%ANDSE%(2)=SE%ANDI=1THENI=2:PRINT"DUE REMI
":GOTO310                                         :rem 253
290 IFI=7ANDSE%=17THENPRINT"UNO SCHELETRO":GOTO310
:rem 190
300 PRINTAR$(I) " "NO$(I)                       :rem 16
310 NEXT:IFOD$(0)="GUARDA"OROD$(0)="OSSERVA"THEN410
:rem 74
320 IFT0%=2THENPRINT"LE POLVERI DELL'ARME= RIA SI SONO AC
CESE!...":PRINT"BUUMMMM!!!":GOTO3050          :rem 103
330 IFSE%=30ANDT0%=1THENT0%=2                  :rem 179
340 IFCR=1ANDSE%(15)<>24THENPRINT"LA STREGA MI HA TRA = S
FORMATO IN UN TOPO!{GIU'}":GOTO3050          :rem 58
350 IFRND(TI)<.9THEN370                          :rem 101
360 IFSE%=15ANDCA%=1THENPRINT"IL CANE MI ATTACCA!!..SONO
MORTO":GOTO3050                                  :rem 78
370 IFSE%=15ANDCA%<>2THENCA%=1:PRINT"IL CANE MI GUARDA MI
= NACCIOSO!!                                     :rem 58

```

```

380 IFSE%=21ANDLEN(NO$(14))<7THENPRINT"I TOPI NON MI FANN
O ANDARE A NORD. :rem 23
390 IFSE%=21ANDSE%(0)=21THENN%=22:NO$(14)="TOPI MORTI
:rem 83
400 IF(SE%<>14ANDSE%<>15ANDSE%<>17ANDSE%<>18ANDSE%<>19AND
SE%<>20)ANDT0%<0THENT0%=0 :rem 114
410 OD$="":INPUT{GIU'}COSA DEVO FARE";OD$:IFOD$=""THENPR
INT"SII PIU'ESPLICITO!!":GOTO410 :rem 65
420 IFSE%=24ANDSE%(22)<>-10ANDSE%(15)<>24ANDCR<1THENCR=1:
PRINT"COMPARE UNA STREGA!! :rem 220
430 IFCR=1ANDSE%(15)=24THENCR=2:PRINT"FUNZIONA!!! LA STRE
GA SCAPPA!!":FORT=0TO750:NEXTT :rem 89
440 FORI=0TO10:OD$(I)="" :NEXT:PU%=0:CO%=0:FORI=1TOLEN(OD$
) :rem 239
450 MI$=MID$(OD$,I,1):IFMI$<>" "ANDMI$<>" "THENOD$(PU%)=O
D$(PU%)+MI$:CO%=0 :rem 110
460 IF(MI$=""ORMI$="" )ANDCO%=0THENCO%=1:PU%=PU%+1
:rem 106
470 NEXT :rem 218
480 IFOD$(0)="VAI"ANDOD$(1)="SOPRA"THENOD$(2)=OD$(3):GOTO
790 :rem 202
490 IFOD$(0)="VAI"ANDLEFT$(OD$(1),2)="SU"THEN790:rem 199
500 IFOD$(0)="VAI"THENOD$(0)=OD$(2) :rem 155
510 IFOD$(0)=LEFT$("NORD",LEN(OD$(0)))ANDN%<>-1ANDSE%=5AN
DO%=-1THENSE%=N%:GOTO860 :rem 244
520 IFOD$(0)=LEFT$("NORD",LEN(OD$(0)))ANDN%<>-1THENSE%=N%
:GOTO90 :rem 220
530 IFOD$(0)=LEFT$("SUD",LEN(OD$(0)))ANDS%<>-1ANDSE%=13AN
DO%=-1THENSE%=S%:GOTO830 :rem 229
540 IFOD$(0)=LEFT$("SUD",LEN(OD$(0)))ANDS%<>-1THENSE%=S%:
GOTO90 :rem 161
550 IFOD$(0)=LEFT$("EST",LEN(OD$(0)))ANDE%<>-1THENSE%=E%:
GOTO90 :rem 134
560 IFOD$(0)=LEFT$("OVEST",LEN(OD$(0)))ANDO%<>-1THENSE%=O
%:GOTO90 :rem 64
570 IFOD$(0)="SALI"THEN790 :rem 220
580 IF(OD$(0)="SCENDI"OROD$(1)="GIU"OROD$(1)="SOTTO")ANDG
I%<>-1THENSE%=GI%:GOTO90 :rem 58
590 IF(OD$(0)="GUARDA"OROD$(0)="*"OROD$(0)="OSSERVA")ANDO
D$(2)=""THEN150 :rem 63
600 IFOD$(0)="PRENDI"OROD$(0)="PIGLIA"OROD$(0)="AFFERRA"
HEN920 :rem 203
610 IFOD$(0)="ACCENDI"THEN1100 :rem 183
620 IFOD$(0)="SPEGNI"THEN1150 :rem 156
630 IFOD$(0)="ROMPI"OROD$(0)="SPACCA"OROD$(2)="SPEZZA"THE
N1190 :rem 160
640 IFOD$(0)="TAGLIA"OROD$(0)="ABBATTI"THEN1290 :rem 224
650 IFOD$(0)=""THENOD$(0)="{SPAZI}" :rem 107
660 IFOD$(0)=LEFT$("INVENTARIO",LEN(OD$(0)))THEN1380
:rem 18

```



```

670 IFOD$(0)="BUTTA"OROD$(0)="POSA"OROD$(0)="GETTA"THEN14
      50                                     :rem 186
680 IFOD$(0)="APRI"THEN1580               :rem 15
690 IFOD$(0)="OSSERVA"OROD$(0)="ESAMINA"OROD$(0)="GUARDA"
      THEN1620                             :rem 104
700 IFOD$(0)="RUOTA"OROD$(0)="GIRA"THEN1680 :rem 229
710 IFOD$(0)="LEGGI"THEN1700              :rem 63
720 IFOD$(0)="ABBASSA"THEN1730            :rem 200
730 IFOD$(0)="NUTRI"OROD$(0)="ALIMENTA"THEN1780 :rem 24
740 IFOD$(0)="UCCIDI"THEN1820            :rem 142
750 IFOD$(0)=LEFT$("AIUTO!!!!!!",LEN(OD$(0)))THEN1960
      :rem 62
760 IFOD$(0)="MI"ANDOD$(1)="ARRENDO"THEN3050 :rem 10
770 IFOD$(0)="SPOSTA"THEN1990             :rem 194
780 PRINT"{GIU'}NON CAPISCO:NON CONO =SCO QUESTE PAROLE O
      { 3 SPAZI}NON POSSO APPLICARLE{ 2 SPAZI}ORA.":GOTO410
      :rem 65
790 IFOD$(2)=""THENPRINT"DOVE SALGO ?":GOTO410 :rem 213
800 IFOD$(2)<>"TORRE"ANDOD$(2)<>"BARCA"THENPRINT"COSA????
      ":GOTO410                             :rem 249
810 IFOD$(2)="TORRE"THEN890               :rem 63
820 IFSE%<>SE%(3)THENPRINT"NON VEDO NESSUNA BARCAQUI.":GO
      TO410                                   :rem 85
830 P0$="SONO SULLA BARCA NELLABAIA.A NORD C'E'IL MA=RE C
      ON L'ISOLA.                            :rem 152
840 SE%(3)=5:N%=-1:S%=-1:E%=-1:O%=-1:SU%=-1:GI%=5:IFSE%(1
      )=-1ANDSE%(2)=-1THENN%=13             :rem 3
850 P1$="" :P2$="" :P3$="" :IFSE%<>13THEN150 :rem 175
860 P0$="SONO SULLA BARCA SULLASPIAGGIA DELL'ISOLA. ASUD
      C'E'IL MARE;OLTRE C'E'LA BAIA.        :rem 138
870 SE%(3)=13:N%=-1:S%=-1:E%=-1:O%=-1:SU%=-1:GI%=13:IFSE%
      (1)=-1ANDSE%(2)=-1THENS%=5          :rem 58
880 GOTO150                                :rem 111
890 IFSE%<>25THENPRINT"QUALE TORRE?":GOTO410 :rem 25
900 IFSE%(16)=-1THENSE%=27:GOTO90         :rem 166
910 PRINT"NON HO NIENTE CHE MI{ 2 SPAZI}PERMETTA DI FARLO
      .":GOTO410                             :rem 175
920 IFNO%>=5THENPRINT"NON POSSO PORTARE NIENTE ALTRO.":GO
      TO410                                   :rem 179
930 IFOD$(2)="REMI"THEN1070               :rem 10
940 IFOD$(2)=""THENPRINTOD$(0)" COSA ?":GOTO410 :rem 140
950 IFOD$(2)="BARCA"OROD$(2)="CANE"OROD$(2)="TRAPPOLA"THE
      NPRINT"NON ESSERE SCIOCCO!":GOTO410   :rem 81
960 IFOD$(2)="TOPO"OROD$(2)="TOPI"OROD$(2)="SCHELETRO"THE
      NPRINT"NON ESSERE SCIOCCO!":GOTO410   :rem 166
970 IFOD$(2)="ALBERO"THENOD$(2)="TRONCO  :rem 60
980 IFOD$(2)="CORONA"THENOD$(2)="CORONA DI PIETRA
      :rem 137
990 IFOD$(2)="TECA"ANDSE%=24THENPRINT"E'BLOCCATA AL SUO P
      IE=DESTALLO!":GOTO410                 :rem 1

```

```

1000 IF (OD$(2) = "CROCEFISSO" OR OD$(2) = "CROCIFISSO") AND SE% = 2
    THEN PRINT "E' FISSATO AL MURO.": GOTO 410           :rem 87
1010 IF OD$(2) = "FANTASMA" OR OD$(2) = "STREGA" THEN PRINT "MA, SEC
    ONDO TE, COME FACCIIO???:": GOTO 410               :rem 145
1020 CO% = 0: FOR I = 0 TO 27                            :rem 202
1030 IF OD$(2) = NO$(I) AND SE%(I) = SE% THEN NO% = NO% + 1: PRINT "O.K.
    ": SE%(I) = -1: CO% = 1: I = 28                    :rem 75
1040 IF BO = 0 AND SE%(12) = -1 THEN PRINT "DIETRO LA BOTTE C'E' UNP
    ASSAGGIO!!": BO = 1: GOSUB 2630                   :rem 26
1050 NEXT: IFCO% = 0 THEN PRINT "NON VEDO "OD$(1) " "OD$(2) " QUI
    .                                                    :rem 1
1060 GOTO 410                                           :rem 149
1070 IF SE%(1) <> SE% OR SE%(2) <> SE% THEN PRINT "NON VEDO I REMI
    QUI.": GOTO 410                                     :rem 153
1080 IF NO% >= 4 THEN PRINT "NON POSSO PORTARLI.": GOTO 410
    :rem 235
1090 PRINT "O.K.": NO% = NO% + 2: SE%(1) = -1: SE%(2) = -1: GOTO 410
    :rem 26
1100 IF OD$(2) = "POLVERI" THEN PRINT "MA SEI PAZZO????": GOTO 410
    0                                                    :rem 52
1110 IF OD$(2) <> "TORCIA" THEN PRINT "COSA????": GOTO 410
    :rem 54
1120 IF SE%(5) <> -1 THEN PRINT "QUALE TORCIA?": GOTO 410
    :rem 239
1130 IF T0% > 0 THEN PRINT "E' GIA' ACCESA.": GOTO 410   :rem 165
1140 PRINT "O.K.": T0% = 1: GOTO 90                    :rem 186
1150 IF OD$(2) <> "TORCIA" THEN PRINT "NON SCHERZARE, PER PIA = CE
    RE!": GOTO 410                                     :rem 20
1160 IF SE%(5) <> -1 THEN PRINT "QUALE TORCIA?": GOTO 410
    :rem 243
1170 IF T0% < 1 THEN PRINT "E' GIA' SPENTA.": GOTO 410  :rem 211
1180 PRINT "O.K.": T0% = 0: GOTO 410                   :rem 233
1190 IF OD$(2) = "CATENA" OR OD$(2) = "CATENE" THEN 1240 :rem 195
1200 IF SE% <> 24 THEN 780                               :rem 189
1210 IF OD$(2) <> "TECA" AND OD$(2) <> "CRISTALLO" THEN PRINT "NON
    SCHERZARE!!": GOTO 410                             :rem 188
1220 IF SE%(4) <> -1 AND SE%(9) <> -1 AND SE%(26) <> -1 THEN 1350
    :rem 121
1230 PRINT "O.K.": NO$(23) = "TECA ROTTA": PRINT "CRASHHH!!": SE
    %(22) = 24: GOTO 410                                :rem 11
1240 IF SE% <> 23 THEN PRINT "QUALI CATENE?": GOTO 410   :rem 101
1250 IF SE%(25) = -10 THEN PRINT "MA SONO GIA' TAGLIATE!": GOTO 4
    10                                                    :rem 208
1260 IF SE%(26) <> -1 THEN 910                           :rem 111
1270 PRINT "O.K.": PRINT "CRRAKKK.": PRINT "IL FANTASMA SE NE
    VA ELIBERA IL PASSAGGIO.                            :rem 179
1280 SE%(25) = -10: FORT = 1 TO 750: NEXT T: GOTO 90    :rem 214
1290 IF OD$(2) = "CATENA" OR OD$(2) = "CATENE" THEN 1240 :rem 196
1300 IF SE% = 26 AND (OD$(2) = "CORDA" OR OD$(2) = "TRAPPOLA") THEN 13
    60                                                    :rem 181

```

```

1310 IFOD$(2) <> "ALBERO" AND OD$(2) <> "TRONCO" THEN PRINT "COSA?
      ??": GOTO 410 :rem 140
1320 IF SE%=7 OR SE%=8 OR SE%=9 OR SE%=10 OR SE%=11 OR SE%=12 THEN 134
      0 :rem 27
1330 PRINT "NON VEDO NESSUN ALBERO QUI.": GOTO 410 :rem 103
1340 IF SE%(9) = -1 THEN PRINT "O.K.": SE%(27) = SE%: GOTO 410
      :rem 226
1350 GOTO 910 :rem 156
1360 IF SE%(9) = -1 OR SE%(6) = -1 OR SE%(26) = -1 THEN PRINT "O.K.": N%
      = 28: S% = 16: SE%(24) = -10: GOTO 410 :rem 142
1370 GOTO 1350 :rem 205
1380 CO% = 0: FOR I = 0 TO 27 :rem 211
1390 IF SE%(I) <> -1 THEN 1430 :rem 130
1400 IF CO% = 0 THEN CO% = 1: PRINT "{RVS}HO LE SEGUENTI COSE:
      :rem 183
1410 IF SE%(1) = -1 AND SE%(2) = -1 AND I = 1 THEN I = 2: PRINT "DUE REMI"
      :GOTO 1430 :rem 159
1420 PRINT AR$(I) " " NO$(I) :rem 68
1430 NEXT: IF CO% = 0 THEN PRINT "NON HO NULLA CON ME.": rem 112
1440 GOTO 410 :rem 151
1450 IF OD$(2) = "CORONA" THEN OD$(2) = "CORONA DI PIETRA
      :rem 178
1460 IF OD$(2) = "" THEN PRINT OD$(0) " COSA ?": GOTO 410: rem 186
1470 IF OD$(2) = "SCALA" AND SE% = 27 THEN PRINT "NON LO FACCIO: POI
      NON POSSO PIU' SCENDERE!!!": GOTO 410 :rem 69
1480 IF OD$(2) = "ALBERO" THEN OD$(2) = "TRONCO :rem 105
1490 IF OD$(2) = "REMI" THEN 1560 :rem 64
1500 CO% = 0: FOR I = 0 TO 27 :rem 205
1510 IF OD$(2) = NO$(I) AND SE%(I) = -1 THEN NO% = NO% - 1: SE%(I) = SE%:
      CO% = 1: I = 28 :rem 79
1520 NEXT: IF CO% = 0 THEN PRINT "NON HO "OD$(2)" CON ME."
      :rem 148
1530 IF CO% = 1 THEN PRINT "O.K. :rem 81
1540 IF OD$(2) = "TRONCO" AND CO% = 1 THEN 130 :rem 164
1550 GOTO 410 :rem 153
1560 IF SE%(2) <> -1 OR SE%(1) <> -1 THEN PRINT "NON HO I REMI CON
      ME.": GOTO 410 :rem 203
1570 PRINT "O.K.": NO% = NO% - 2: SE%(1) = SE%: SE%(2) = SE%: GOTO 410
      :rem 221
1580 IF OD$(2) = "PONTE" AND (SE% = 28 OR SE% = 29) THEN 1730: rem 38
1590 IF OD$(2) <> "PORTA" THEN PRINT "COSA????": GOTO 780
      :rem 16
1600 IF SE%(13) <> -1 THEN 1350 :rem 152
1610 PRINT "NON HO LA CHIAVE GIU =STA.": GOTO 410 :rem 152
      :rem 0

```

Guerra spaziale

I biechi Mingoniani sono in lotta con la Terra per il predominio sul Sistema Solare e tu, al comando della tua astronave da caccia, ingaggi un duello mortale con una nave nemica. Usa il tuo fedele joystick per cambiare direzione e per azionare il cannone laser a lungo raggio. La battaglia può aver luogo in tre ambienti diversi: all'interno del Sistema Solare, oppure nella Zona dei Buchi Neri, oppure infine nello Spazio Esterno: ciò che distingue i tre scenari è la forza di gravità. Nel primo devi stare attento a non passare troppo vicino al Sole, altrimenti la sua attrazione può catturarti; nel secondo c'è un Buco Nero (invisibile, ovviamente!) che devi evitare, e che puoi individuare notando come, passando nella sua zona, la tua rotta venga influenzata dalla sua attrazione; nel terzo, infine, non c'è gravità. La tua nave viene distrutta se

un Mingoniano ti colpisce, se ti scontri con un asteroide o col nemico, oppure infine se cadi nel Sole o in un Buco Nero. Premendo la barra spaziatrice compi un balzo nell'iperspazio e riappari in un punto a caso dello schermo. Quanto ti trovi nel Sistema Solare ricordati che, se il Sole viene spento dai cannoni laser, esso si trasforma in un Buco Nero. Vinci se disintegri col tuo cannone 5 navi nemiche prima di essere a tua volta abbattuto 5 volte. Puoi scegliere tra 9 livelli di difficoltà, che determinano la quantità di asteroidi presenti.

Il gioco, che gira su VIC 20 con espansione da 3K, è diviso in due parti; la seconda deve venir salvata, chiamandola "G", subito dopo la prima. Chi usa il disk drive elimini la parola REM nella riga 622 del programma 1. Sia da nastro che da disco il caricamento del programma 2 avviene automaticamente.

Programma 1

```

10 REM GUERRA SPAZIALE 1                               :rem 149
40 POKE36879,27:PRINT"{CLR}{ 2 SPAZI}{RVS} GUERRA    :rem 8
   { 2 SPAZI}SPAZIALE "
50 PRINT"{ 4 GIU'}UN ATTIMO PREGO..."                :rem 242
60 X=PEEK(56)-2:POKE56,X:POKE52,X                      :rem 231
70 POKE51,PEEK(55):CLR                                   :rem 35
80 CS=256*PEEK(52)+PEEK(51):S=7680                      :rem 133
90 FORI=CSTOCS+511:POKEI,PEEK(I+32768-CS):NEXT          :rem 117
100 READX:IFX<0THEN130                                   :rem 85
110 FORI=XTOX+7:READJ:POKEI,J:NEXT                       :rem 245
120 GOTO100                                              :rem 93

```

```

130 INPUT"{ 2 GIU'}{ 3 SPAZI}ISTRUZIONI{ 3 SPAZI}N
    { 3 SIN}";A$                               :rem 111
140 IFA$="N"THEN620                             :rem 31
150 PRINT"{CLR}DEVI COMBATTERE UNA"           :rem 236
160 PRINT"BATTAGLIA MORTALE CON-"             :rem 18
170 PRINT"{SU}TRO UNA NAVE MINGO-"           :rem 164
180 PRINT"NIANA NELLA FASCIA DI"             :rem 113
190 PRINT"ASTEROIDI DI ANTARES."            :rem 216
200 PRINT"LA TUA NAVE E' VIOLA,"            :rem 23
210 PRINT"LA NAVE AVVERSARIA"               :rem 15
220 PRINT"BLU. USA IL JOYSTICK"              :rem 100
230 PRINT"PER MUOVERTI E SPARA-"            :rem 177
240 PRINT"RE CON IL LASER. CON"             :rem 248
250 PRINT"'SPAZIO' SALTI NEL-"              :rem 21
260 PRINT"L'IPERSPAZIO."                    :rem 16
310 PRINT"{ 2 GIU'}PREMI UN TASTO"           :rem 50
320 A$="":GETA$:IFA$=""THEN320               :rem 109
330 PRINT"{CLR}VIENI SCONFITTO SE:"         :rem 0
340 PRINT"{GIU'}1 SEI COLPITO DAI"          :rem 115
350 PRINT"{ 2 SPAZI}MINGONIANI"             :rem 82
360 PRINT"{GIU'}2 SEI RISUCCHIATO IN"       :rem 99
370 PRINT"{ 2 SPAZI}UN BUCO NERO"           :rem 107
380 PRINT"{GIU'}3 COLLIDI CON UN"           :rem 51
390 PRINT"{ 2 SPAZI}ASTEROIDE"              :rem 13
400 PRINT"{GIU'}4 CADI NEL SOLE PER"        :rem 180
410 PRINT"{ 2 SPAZI}EFFETTO DELLA"          :rem 213
420 PRINT"{ 2 SPAZI}GRAVITA'"               :rem 156
430 PRINT"{GIU'}5 TI SCONTRI CON LA"        :rem 218
440 PRINT"{ 2 SPAZI}NAVE MINGONIANA"        :rem 116
450 PRINT"{ 2 GIU'}PREMI UN TASTO"          :rem 55
460 A$="":GETA$:IFA$=""THEN460              :rem 119
620 PRINT"{CLR}{ 3 GIU'}L"CHR$(34)"G"CHR$(34); :rem 130
622 REM PRINT",8"                             :rem 179
625 PRINT"{HOME}":POKE631,13:POKE632,82:POKE633,213:POKE6
    34,13:POKE198,4:END                       :rem 135
630 DATA7168,192,240,127,103,34,54,63,50   :rem 231
640 DATA7176,4,14,62,227,227,62,14,4       :rem 33
650 DATA7184,50,63,54,34,103,127,240,192   :rem 231
660 DATA7192,24,24,60,36,102,231,126,24    :rem 172
670 DATA7208,24,126,231,102,36,60,24,24    :rem 171
680 DATA7216,3,15,254,230,68,108,252,76    :rem 190
690 DATA7224,32,112,124,199,199,124,112,32 :rem 78
700 DATA7232,76,252,108,68,230,254,15,3    :rem 181
710 DATA7240,0,0,12,28,56,48,0,0           :rem 67
720 DATA7248,0,0,48,56,28,12,0,0           :rem 76
730 DATA7256,0,24,24,24,24,24,0,0         :rem 118
740 DATA7264,0,0,0,62,62,0,0,0             :rem 216
750 DATA7296,0,112,126,102,32,48,48,0      :rem 75
760 DATA7304,228,18,37,68,36,18,33,198     :rem 156
770 DATA7312,0,48,48,32,102,126,112,0     :rem 66
780 DATA7320,68,170,145,0,34,85,137,129    :rem 192

```


790	DATA7328,153,90,60,255,255,60,90,153	:rem	246
800	DATA7336,129,137,85,34,0,145,170,68	:rem	192
810	DATA7344,0,14,126,102,4,12,12,0	:rem	208
820	DATA7352,198,33,18,36,68,37,18,228	:rem	156
830	DATA7360,0,12,12,4,102,126,14,0,-1	:rem	90

Programma 2

1	REM GUERRA SPAZIALE 2	:rem	102
3	PRINT"{CLR}DIFFICOLTA"	:rem	147
5	PRINT"{GIU'}DA{ 2 SPAZI}{RVS}1{OFF}-FACILISSIMO"	:rem	209
7	PRINT"A{ 3 SPAZI}{RVS}9{OFF}-IMPOSSIBILE	:rem	113
9	A\$="":GETA\$:SL=VAL(A\$):IFA\$<"1"ORAS>"9"THEN9	:rem	85
11	PRINT"{CLR}OPZIONI"	:rem	238
13	PRINT"{RVS}S{OFF}ISTEMA SOLARE"	:rem	181
15	PRINT"{RVS}B{OFF}UCHI NERI"	:rem	116
17	PRINT"{RVS}N{OFF}IENTE GRAVITA'"	:rem	213
19	GETA\$:IFA\$<>"S"ANDA\$<>"B"ANDA\$<>"N"THEN19	:rem	3
20	IFA\$="S"THENG=1	:rem	14
21	IFA\$="B"THENG=2	:rem	255
23	IFA\$="N"THENG=0	:rem	11
40	C=22:R=23:S=7680:A=30720:DD=37154	:rem	204
50	P1=37151:P2=37152	:rem	31
60	POKE36879,27:T=3:CY=4:CE=6:CA=0	:rem	224
70	CS=7:CQ=0:CM=0:X=-1	:rem	16
80	V=36878:S1=V-2:S2=V-1:DIMC%(2,2),OC%(2,2)	:rem	140
90	FORI=0TO2:FORJJ=0TO2:X=X+1	:rem	85
100	DC%(I,JJ)=X:OC%(I,JJ)=X+16:NEXT:NEXT	:rem	144
110	DEFFNA(Z)=S+X+C*Y	:rem	229
120	DEFFNB(Z)=PEEK(FNA(Z))	:rem	22
130	DEFFNR(Z)=INT(RND(1)*Z)	:rem	87
140	POKEV-9,255:PRINT"{CLR}{BLU}":X=S+A	:rem	147
150	FORI=XTOX+505:POKEX,T:NEXT	:rem	213
160	POKE36879,62:FORI=1TO3+2*SL	:rem	63
170	X=FNR(C):Y=FNR(R)	:rem	0
180	POKEFNA(0),46:POKEFNA(0)+A,CA:NEXT	:rem	8
190	IFG<>1THEN220	:rem	222
200	X=11:Y=12:POKEFNA(0),20	:rem	9
210	POKEFNA(0)+A,CS:SX=X:SY=Y	:rem	60
220	IFG=2THENSX=FNR(C):SY=FNR(R)	:rem	22
230	X=FNR(C):Y=FNR(R):IFFNB(0)<>32THEN230	:rem	192
240	D=X:E=Y:Q=FNA(0):U=-1:O=FNR(3)-1	:rem	90
250	POKEFNA(0),DC%(U+1,O+1):POKEFNA(0)+A,CY	:rem	192
260	X=FNR(C):Y=FNR(R):IFFNB(0)<>32THEN260	:rem	198
270	H=X:L=Y:K=FNA(0):M=1:N=FNR(3)-1	:rem	44
280	POKEFNA(0),DC%(M+1,N+1):POKEFNA(0)+A,CE	:rem	166
290	A\$="":GETA\$:IFA\$=" "THEND=FNR(C):E=FNR(R)	:rem	30
300	GOSUB1010:IFFBTHENX=D:Y=E:PX=U:PY=O:GOSUB680:rem	126	

310	B=0:F=0:IFJ0THENB=1	:rem	82
320	IFJ2THENB=-1	:rem	172
330	IFJ1THENF=1	:rem	131
340	IFJ3THENF=-1	:rem	179
350	IFB=0ANDF=0THENB=U:F=0	:rem	107
360	U=B:O=F:IFG=0THEN410	:rem	186
370	B=SX-D:F=SY-E:J=SQR(B*B+F*F)	:rem	167
380	J=(20-J)/30:J=1-J*J	:rem	187
390	IFRND(1)<JTHEN410	:rem	219
400	B=SGN(B):F=SGN(F):D=D+B:E=E+F:GOTO420	:rem	177
410	D=D+U:E=E+O	:rem	85
420	IFE<0THENE=R	:rem	217
430	IFE>RTHENE=0	:rem	220
440	IFD>CTHEND=0	:rem	204
450	IFD<0THEND=C	:rem	203
460	X=D:Y=E:J=FNB(0):IFJ=32THEN480	:rem	32
470	IFJ=46ORJ=20ORFNA(0)=KTHEN820	:rem	237
480	IFNOT(X=SXANDY=SY)THEN510	:rem	135
490	A\$="SEI STATO INGOIATO DA UN BUCO NERO!"	:rem	239
500	WC=WC+1:GOTO970	:rem	117
510	POKEQ,32:POKEQ+A,T:Q=FNA(0)	:rem	107
520	POKEQ,DC%(U+1,O+1):POKEQ+A,CY	:rem	182
530	J=0:IFFNR(9)>SLTHEN560	:rem	47
540	M=D-H:N=E-L:M=SGN(M):N=SGN(N)	:rem	245
550	IFM=0ORN=0THENJ=1	:rem	38
560	H=H+M:L=L+N:IFH<0THENH=C	:rem	220
570	IFH>CTHENH=0	:rem	216
580	IFL>RTHENL=0	:rem	240
590	IFL<0THENL=R	:rem	239
600	X=H:Y=L	:rem	143
610	IFFNB(0)=32THEN630	:rem	231
620	M=FNR(3)-1:N=FNR(3)-1:GOTO560	:rem	133
630	POKEK,32:POKEK+A,T:K=FNA(0)	:rem	92
640	POKEK,DC%(M+1,N+1):POKEK+A,CE	:rem	144
650	IFNOT(RND(1)<.1OR(J=1ANDFNR(9)<SL))THEN670	:rem	1
660	PX=M:PY=N:GOSUB680	:rem	148
670	GOTO290	:rem	113
680	Z=PX*PY:POKEV,8:IFZ=1THENJ=10	:rem	123
690	IFZ=-1THENJ=9	:rem	18
700	IFZ=0ANDPX=0THENJ=11	:rem	237
710	IFZ=0ANDPY=0THENJ=12	:rem	240
720	FORI=1TO10:X=X+PX:Y=Y+PY:POKES2,230-I	:rem	87
730	IFI<>1THENPOKEZ,32:POKEZ+A,T:IFX>CTHENX=0	:rem	171
740	IFX<0THENX=C	:rem	245
750	IFY>RTHENY=0	:rem	9
760	IFY<0THENY=R	:rem	8
770	B=FNB(0):IFB=32THEN800	:rem	232
780	IFNOT(B=46ORB=20ORFNA(0)=KORFNA(0)=Q)THEN800	:rem	166
790	I=10:NEXT:GOTO820	:rem	13
800	Z=FNA(0):POKEZ,J:POKEZ+A,CM:NEXT	:rem	34
810	POKEZ,32:POKEZ+A,T:POKEV,0:RETURN	:rem	151

```

820 POKES2,230:SC=X-1:IFSC<0THENSC=0           :rem 207
830 FC=X+1:IFFC>CTHENFC=C                       :rem 32
840 SR=Y-1:IFSR<0THENSR=0                       :rem 80
850 FR=Y+1:IFFR>RTHENFR=R                       :rem 110
860 FORX=SCTOFC:FORY=SRTOFR:J=OC%(X-SC,Y-SR)   :rem 174
870 POKEFNA(0),J:POKEFNA(0)+A,CQ:NEXT:NEXT      :rem 119
880 POKES1,220:FORJJ=15TO0STEP-1                :rem 216
890 POKEV,JJ:FORJ1=1TO50:NEXT:NEXT              :rem 234
900 POKEV,0:FORS=SCTOFC:FORY=SRTOFR            :rem 137
910 POKEFNA(0),32:POKEFNA(0)+A,T:NEXT:NEXT     :rem 77
920 IFPEEK(Q)<>32THEN940                          :rem 156
930 A$="TI SEI DISINTEGRATO!":WC=WC+1:GOTO970   :rem 200
940 IFPEEK(K)<>32THEN960                          :rem 154
950 A$="HAI VINTO!":WH=WH+1:GOTO970             :rem 43
960 GOTO810                                       :rem 113
970 POKEV-9,240:PRINT"{CLR}{ 3 GIU' }";A$      :rem 190
980 PRINT"{ 2 GIU' }PUNTI: VIC";WC;"{ 2 SPAZI }TU";WH :rem 23
990 FORJJ=1TO2500:NEXT:GOTO1050                 :rem 167
1000 END                                           :rem 152
1010 POKEDD,127:P=PEEK(P2)AND128                :rem 108
1020 J0=- (P=0):POKEDD,255                       :rem 110
1030 P=PEEK(P1):J1=- ((PAND8)=0):J2=- ((PAND16)=0) :rem 138
1040 J3=- ((PAND4)=0):FB=- ((PAND32)=0):RETURN  :rem 239
1050 IFWH=5THEN1080                              :rem 94
1060 IFWC=5THEN1100                              :rem 83
1070 GOTO140                                       :rem 150
1080 PRINT"{CLR}{ 3 SPAZI }CONGRATULAZIONI!"    :rem 199
1090 GOTO1110                                       :rem 198
1100 PRINT"{CLR}{ 3 SPAZI }SEI UN DISASTRO!"    :rem 52
1110 FORJJ=1TO4000:NEXT:RUN                       :rem 187

```

ERRATA CORRIGE

Nel programma "RUBRICA INDIRIZZI" per C16, pubblicato sul n. 23, vanno apportate alcune correzioni:

- cancellare le linee 1970 e 2000 e la linea 2120
- modificare le seguenti linee:

```

1235 IFG$=CHR$(13) THENPOKE239,0:GOTO1270      :rem 77
1345 IFG$=CHR$(13) THENPOKE239,0:GOTO1470      :rem 81
1425 IFG$=CHR$(13) THENPOKE239,0:GOTO1460      :rem 79
1960 INPUT#7,Q:FORSS=1TOQ:FORJ=1TO6:INPUT#7,A$(SS,J):NEXT :rem 38
      :NEXT
2110 PRINT#6,Q:FORLS=1TOQ:FORJ=1TO6:PRINT#6,A$(LS,J)+CHR$(1):NEXT:NEXT :rem 178
2880 IFG$="{GIU' }"ORG$="{DES }"ORG$="{SU }"ORG$="{SIN }"ORG$ :rem 104
      =CHR$(20) THENG$=""

```

Titolatore di cassette

Stampante

Se possedete una stampante Commodore e, come me, molte decine di cassette contenenti registrazioni musicali (o anche software per il vostro C16), allora questo semplice programma può esservi molto utile. Esso vi permette di stampare le copertine dei vostri nastri in modo razionale e ordinato.

Il funzionamento del programma si basa sul fatto che, sulle stampanti Commodore, una matrice di 25 righe per 40 colonne ha esattamente le stesse dimensioni di una copertina standard per cassette. Come ricorderete, anche una "schermata" del vostro C16 è composta da 25 righe di 40 caratteri; il programma mostra quindi sul video la copertina (ancora priva di titoli e nomi), in modo che voi possiate compilarla. A tale scopo compare (in alto a sinistra) un cursore, che potete spostare coi soliti tasti di controllo. Potete digitare ciò che volete, eventualmente anche cancellando le parti precompilate. Attenzione però: il tasto INST/DEL non funziona, per cui qualsiasi correzione va

effettuata posizionandosi sul carattere errato e battendo quello giusto.

Quando la copertina è pronta, potete stamparla premendo il tasto RETURN. Effettuata la stampa, potete ottenere un'altra copia premendo nuovamente RETURN (eventualmente dopo aver fatto delle correzioni), oppure creare una nuova copertina usando il tasto ESC. Battete infine INST/DEL quando avete finito di utilizzare il programma.

Per concludere, vi segnaliamo la versatilità della subroutine che stampa la copertina (linee 10000-10070). Essa effettua un "hard copy" in bassa risoluzione del video, cioè stampa su carta la copia esatta di una schermata di 1000 caratteri del C16. L'unica limitazione è che un carattere in reverse sul video appare normale su carta. Potete inserire la routine in un vostro programma, e chiamarla con un GOSUB10000 ogni volta che volete stampare il contenuto dello schermo.

```

1 X=1:S=0                                     :rem 241
2 COLOR0,1:COLOR4,2,2:PRINT" [<1>]"         :rem 115
3 GOSUB100                                     :rem 68
10 GETA$                                     :rem 166
11 P=3071+X:C=PEEK(P):POKEP,102:POKEP,C      :rem 93
15 IFA$=""THEN10                              :rem 107
20 A=ASC(A$)                                  :rem 109
21 IFA=13THENGOSUB10000:GOTO10              :rem 72
22 IFA=27THENRUN                             :rem 254

```

```

23 IFA=20THENSCNCLR:END :rem 217
24 IFA=145THENX=X-40:S=1 :rem 181
25 IFA=157THENX=X-1:S=1 :rem 134
26 IFA=13THENX=X+40:S=1 :rem 127
30 IFA=17THENX=X+40:S=1 :rem 126
35 IFA=29THENX=X+1:S=1 :rem 83
37 IFX<1ORX>1000THENX=1 :rem 171
40 IFS>0THENS=0:GOTO10 :rem 119
60 IFA>63ANDA<96THENA=A-64 :rem 33
62 IFA>31ANDA<64THEN72 :rem 44
64 IFA>96ANDA<128THENA=A-32 :rem 82
66 IFA>159ANDA<192THENA=A-64 :rem 138
72 IFX<1ORX>1000THENX=1 :rem 170
73 P=3071+X :rem 69
75 POKEP,A :rem 88
76 X=X+1 :rem 182
80 GOTO10 :rem 2
100 PRINT"[<A>]{ 19 *}[<R>]{ 18 *}[<S>]"; :rem 109
110 PRINT"----- A ----- B -----"; :rem 43
120 PRINT"- 1){ 16 SPAZI}- 1){ 15 SPAZI}-"; :rem 234
130 PRINT"- 2){ 16 SPAZI}- 2){ 15 SPAZI}-"; :rem 237
135 PRINT"- 3){ 16 SPAZI}- 3){ 15 SPAZI}-"; :rem 244
140 PRINT"- 4){ 16 SPAZI}- 4){ 15 SPAZI}-"; :rem 242
145 PRINT"- 5){ 16 SPAZI}- 5){ 15 SPAZI}-"; :rem 249
150 PRINT"- 6){ 16 SPAZI}- 6){ 15 SPAZI}-"; :rem 247
160 PRINT"- 7){ 16 SPAZI}- 7){ 15 SPAZI}-"; :rem 250
165 PRINT"- 8){ 16 SPAZI}- 8){ 15 SPAZI}-"; :rem 1
170 PRINT"- 9){ 16 SPAZI}- 9){ 15 SPAZI}-"; :rem 255
175 PRINT"-10){ 16 SPAZI}-10){ 15 SPAZI}-"; :rem 84
185 PRINT"-11){ 16 SPAZI}-11){ 15 SPAZI}-"; :rem 87
190 PRINT"-12){ 16 SPAZI}-12){ 15 SPAZI}-"; :rem 85
200 PRINT"-13){ 16 SPAZI}-13){ 15 SPAZI}-"; :rem 79
210 PRINT" [<Q>]{ 5 *}[<R>]{ 13 *}[<E>]{ 18 *}[<W>]"; :rem 96
220 PRINT"-N.{ 3 SPAZI}-LATO A:{ 25 SPAZI}-"; :rem 94
225 PRINT"-TIPO -LATO B:{ 25 SPAZI}-"; :rem 36
230 PRINT" [<Q>]{ 5 *}[<E>]{ 13 *}[<R>]{ 18 *}[<W>]"; :rem 98
240 PRINT"-14){ 16 SPAZI}-14){ 15 SPAZI}-"; :rem 85
250 PRINT"-15){ 16 SPAZI}-15){ 15 SPAZI}-"; :rem 88
260 PRINT"-16){ 16 SPAZI}-16){ 15 SPAZI}-"; :rem 91
270 PRINT"-17){ 16 SPAZI}-17){ 15 SPAZI}-"; :rem 94
280 PRINT"-18){ 16 SPAZI}-18){ 15 SPAZI}-"; :rem 97
290 PRINT" [<Z>]{ 19 *}[<E>]{ 18 *]"; :rem 197
300 POKE4071,125 :rem 82
399 RETURN :rem 133
10000 AZ$="":OPEN4,4:LZ=3072:FORLZ=LZTO4071STEP40:FORIZ=L
ZTOLZ+39:ZZ=PEEK(IZ) :rem 158
10010 IFZZ<96ANDZZ>63THENZZ=ZZ+32:GOTO10040 :rem 218
10020 IFZZ<32THENZZ=ZZ+64:GOTO10040 :rem 168

```



```

10030 IFZZ<128ANDZZ>96THENZZ=ZZ+64           :rem 171
10040 ZZ$=CHR$(ZZ):AZ$=AZ$+ZZ$:NEXT          :rem 129
10050 IFAZ$="{ 40 SPAZI}"THENPRINT#4:GOTO10070 :rem 67
10060 PRINT#4,AZ$                               :rem 198
10070 AZ$="":NEXT:CLOSE4:RETURN               :rem 175

```

C16~~+~~-plus/4

Explorer

Questo simpatico gioco spaziale gira su C16 e PLUS/4.

In esso bisogna guidare una navetta spaziale attraverso lo spazio, superando vari ostacoli.

Dopo la partenza bisogna sorvolare la città a minor altezza possibile, cercando di non scontrarsi contro i palazzi.

In questa prima parte del viaggio più si vola in basso, più il punteggio aumenta.

Se invece si vola alla massima altezza concessa il punteggio decresce. Dopo aver superato la città bisognerà evitare le numerose meteore presenti attorno al pianeta.

In questa fase, al contrario della precedente, l'altezza di volo non conta e il punteggio aumenta sempre allo stesso ritmo.

Naturalmente c'è il rischio di schiantarsi contro le meteore e i

palazzi, e in tal caso, per ogni scontro, il punteggio diminuisce di 100 punti.

Dopo aver superato tutte e due le zone il computer mostra il punteggio finale. Per muovere l'astronave in alto e in basso sono necessari i tasti:

Q..... per alzarsi e

Z..... per abbassarsi.

Il punteggio è sempre aggiornato e viene mostrato nell'angolo in alto a sinistra dello schermo.

N.B.

Il programma utilizza una piccola routine in linguaggio macchina, perciò è necessario batterlo con molta cautela e salvarlo prima di farlo girare, perché il minimo errore di battitura potrebbe mandare in Tilt il C16, annullando tutto il lavoro di digitazione.

```

100 POKE55,16:POKE56,39:CLR                   :rem 68
110 GOSUB580                                   :rem 175
120 T$="{RVS} {OFF}{SIN}{SU}{SIN}{SU}{SIN}{SU}{SIN}{SU}
   {SIN}{SU}{SIN}{SU}{SIN}{RVS} {OFF}"      :rem 133
130 COLOR0,7,5:COLOR4,7,5                    :rem 184
140 K$="{RVS} {OFF}{SU}{SIN}":FORI=1TO19:I$=I$+K$:NEXT
                                                :rem 160

```

```

150 SA$="{ 5 SPAZI}{GIU'}{ 5 SIN} N[<Y>]M {GIU'}{ 5 SIN}
      M[<P>]N {GIU'}{ 5 SIN}{ 3 SPAZI}" :rem 209
160 RR$="{HOME}":FORI=1TO24:RR$=RR$+"{GIU'}":NEXT:FORI=1T
      039:CC$=CC$+"{DES}":NEXT :rem 139
170 W$="{HOME}"+LEFT$(RR$,25)+LEFT$(CC$,38) :rem 217
180 FORI=11000TO11071:READA:POKEI,A:NEXT :rem 196
190 PRINT"{CLR}";:K1=11000 :rem 30
200 P=198:R=12:C=20:SA=3072:SL=2048:LM=300 :rem 103
210 PO=SA+(R*40)+C :rem 205
220 PRINT"{CLR}";TAB(16);"PARTE UNO"; :rem 234
230 FORSC=1TO300:SYSK1:PRINTW$;LEFT$(I$,RND(1)*70);
      :rem 83
240 IFPEEK(PO+83)=160ORPEEK(PO+82)=160THENGOSUB690
      :rem 17
250 A=PEEK(P):IFA=62THENGOSUB450 :rem 210
260 IFA=12THENGOSUB470 :rem 82
270 S1=S1+R-6:PRINT"{HOME}{RVS}";:PRINTS1;:PRINT"{OFF}
      {SIN}{ 2 SPAZI}"; :rem 36
280 PRINTLEFT$(RR$,R);LEFT$(CC$,C);SA$;:NEXT :rem 168
290 PRINT"{HOME}";TAB(16);"PARTE DUE"; :rem 93
300 FORSC=1TO300:SYSK1 :rem 105
310 IFSW=0THENSW=1:GOTO350 :rem 140
320 SW=0 :rem 172
330 HK=INT(RND(1)*22):IFHK<5THEN330 :rem 40
340 FK=3072+(HK*40)+39:POKEFK,81 :rem 244
350 IFPEEK(PO+3)=160ORPEEK(PO+44)=81ORPEEK(PO+4)=81THENGO
      SUB690 :rem 102
360 A=PEEK(P):IFA=62THENGOSUB450 :rem 212
370 IFA=12THENGOSUB470 :rem 84
380 S1=S1+1:PRINT"{HOME}{RVS}";:PRINTS1;:PRINT"{OFF}{SIN}
      { 2 SPAZI}"; :rem 162
390 PRINTLEFT$(RR$,R);LEFT$(CC$,C);SA$;:NEXT:PRINT"{CLR}P
      UNTEGGIO: ";S1 :rem 243
400 FORT=1TO1000:NEXT :rem 26
410 PRINT"{ 4 GIU'}{ 7 SPAZI}VUOI GIOCARE ANCORA (S/N)?" :
      POKE239,0 :rem 94
420 GETKEYA$ :rem 196
430 IFA$<>"S"THENEND :rem 162
440 RUN :rem 141
450 IFR=5THENRETURN :rem 251
460 R=R-1:PO=SA+(R*40)+C:RETURN :rem 103
470 IFR=22THENRETURN :rem 44
480 R=R+1:PO=SA+(R*40)+C:RETURN :rem 103
490 REM DATI PER L.M. :rem 127
500 REM ===== :rem 146
510 DATA173,11,255,201,204,208,249,160,0,162,199,185,201,
      12,153,200,12,185 :rem 90
520 DATA145,13,153,144,13,185,89,14,153,88,14,185
      :rem 169
530 DATA33,15,153,32,15,200,202,208,228,160,0,162
      :rem 130

```

```

540 DATA5,142,238,46,169,32,153,239,12           :rem 133
550 DATA153,183,13,153,127,14,153,71,15,152,24,105,40,168
    ,174,238,46,202                               :rem 6
560 DATA208,228,96                               :rem 178
570 NEXT                                           :rem 219
580 PRINT"{CLR}";TAB(16);"{RVS}EXPLORER{OFF}";    :rem 154
590 PRINT:PRINT:PRINT"GUIDA LA NAVE SPAZIALE SULLE MONTAG
    NE.":PRINT                                     :rem 74
600 PRINT"PIU' VOLI BASSO PIU' PUNTI TOTALIZZI.":PRINT
    :rem 146
610 PRINT"SE VOLI ALL'ALTEZZA MASSIMA PERDI PUNTI.":PRINT
    :rem 89
620 PRINT"NON SCHIANTARTI SULLE MONTAGNE, DOPO LE":PRINT
    :rem 34
630 PRINT"QUALI DOVRAI AFFRONTARE UNA FITTA CADUTA":PRINT
    :rem 104
640 PRINT"DI METEORE.":PRINT                      :rem 254
650 PRINT"OGNI SCONTRO TI FA PERDERE 100 PUNTI !!!":PRINT
    :rem 55
660 PRINTTAB(8);"PREMI UN TASTO PER PARTIRE";    :rem 236
670 GETA$:IFA$=""THEN670                          :rem 93
680 RETURN                                         :rem 126
690 FORI=1TO15:FORJ=1TO99:NEXT:COLOR4,I,3:NEXT:S1=S1-100:
    RETURN                                         :rem 62

```

Decidere



Quello che vi presentiamo è un adattamento al C64 di un vecchio programma scritto per il glorioso PET 2001, capostipite di tutti gli home-computer Commodore. Come si può capire dal titolo, questo programma serve ad aiutarvi nel prendere decisioni. Appena date il RUN, infatti, viene presentata la schermata iniziale con un menù comprendente tre opzioni. La prima (scegliere una tra varie alternative) consente di prendere decisioni su qualcosa di specifico, come ad esempio dove andare in vacan-

za; la seconda (scegliere una serie di azioni tra varie possibilità) permette di decidere che cosa fare, per esempio se fare una passeggiata, andare in piscina, restare a casa, ecc.; la terza, infine, serve a prendere la più classica delle decisioni, vale a dire sì o no. Nei primi due casi sarà necessario innanzitutto introdurre le varie alternative possibili; in seguito, qualsiasi opzione si sia selezionata, bisogna elencare i vari fattori che influenzano la decisione, indicando quale ritenete il principale ed assegnan-

do di conseguenza l'importanza dei fattori rimanenti. Infine i vari fattori vengono considerati uno ad uno per ogni decisione possibile; termina così la fase di introduzione dei dati. A questo punto il computer, dopo pochi secondi di elaborazione, visualizza i risultati: alla decisione migliore viene assegnato il valore 100, mentre le altre seguono

con punteggi più bassi assegnati in base ai fattori introdotti. Naturalmente non pretendiamo certo di fornirvi uno strumento per prendere le decisioni più importanti della vostra vita; tuttavia scoprirete che questo programma può esservi utile, se non altro per divertirvi a "misurare" la coerenza con cui fate le vostre scelte.

```

3 REM -DECIDERE- BY CHRISTIAN ANTONIUTTI           :rem 245
10 POKE53280,0:POKE53281,0:PRINT"{RED}"           :rem 107
20 PRINT"{CLR}{ 4 GIU' }{ 15 DES }{RVS}MAGICSOFT" :rem 108
30 PRINT"{ 3 GIU' }{GRN}{ 2 DES }{ 3 Q }{ 2 SPAZI }{ 4 Q }
   { 2 SPAZI }{ 3 Q }{SPAZI}Q{SPAZI}{ 3 Q }{ 2 SPAZI }
   { 4 Q }{SPAZI}{ 3 Q }{ 2 SPAZI }{ 4 Q }"       :rem 8
40 PRINT"{ 2 DES }Q{ 2 SPAZI }Q{SPAZI}Q{ 3 SPAZI }{SPAZI}Q
   { 3 SPAZI }{SPAZI}Q{SPAZI}Q{ 2 SPAZI }Q{SPAZI}Q
   { 3 SPAZI }{SPAZI}Q{ 2 SPAZI }Q{SPAZI}Q{ 3 SPAZI }"
                                                    :rem 10
50 PRINT"{ 2 DES }Q{ 2 SPAZI }Q{SPAZI}{ 2 Q }{ 2 SPAZI }
   {SPAZI}Q{ 3 SPAZI }{SPAZI}Q{SPAZI}Q{ 2 SPAZI }Q{SPAZI }
   { 2 Q }{ 2 SPAZI }{SPAZI}{ 3 Q } {SPAZI}{ 2 Q }{ 2 SPAZI }
   "                                                    :rem 15
60 PRINT"{ 2 DES }Q{ 2 SPAZI }Q{SPAZI}Q{ 3 SPAZI }{SPAZI}Q
   { 3 SPAZI }{SPAZI}Q{SPAZI}Q{ 2 SPAZI }Q{SPAZI}Q
   { 3 SPAZI }{SPAZI}Q Q{ 2 SPAZI }Q{ 3 SPAZI }"     :rem 108
70 PRINT"{ 2 DES }{ 3 Q }{ 2 SPAZI }{ 4 Q }{ 2 SPAZI }{ 3 Q }
   {SPAZI}Q{SPAZI}{ 3 Q }{ 2 SPAZI }{ 4 Q }{SPAZI}Q
   { 2 SPAZI }Q{SPAZI}{ 4 Q }"                       :rem 74
80 PRINT"{ 5 GIU' }{WHT}{ 8 DES }BY CHRISTIAN ANTONIUTTI"
                                                    :rem 202
90 FOR T=1 TO 3000:NEXT:PRINT"{CLR}{N}":POKE53280,1:POKE53281,1
                                                    :rem 46
160 DIM L$(10),F$(10),V(10),C(10,10),D(10),Z(10)   :rem 23
180 E$="FINE"                                         :rem 165
190 FOR J=1 TO 10:GETR$:NEXT                         :rem 70
200 GOSUB 2000                                       :rem 212
210 PRINT"POSSO AIUTARTI A PRENDERE UNA DECISIONE.";
                                                    :rem 81
220 PRINT"PRIMA DEVO PORTI ALCUNE DOMANDE QUINDI""
                                                    :rem 182
230 PRINT"ANALIZZERO' LE INFORMAZIONI CHE MI DAI.":PRINT
                                                    :rem 219
240 PRINT"{ 5 SPAZI }-----":PRINT
                                                    :rem 26
250 PRINT"QUAL E' LA MIGLIOR DESCRIZIONE DEL TIPO"
                                                    :rem 119

```

```

260 PRINT"DI DECISIONE CHE DEVI PRENDERE?":PRINT:rem 220
270 PRINT"1. SCEGLIERE UNA TRA VARIE ALTERNATIVE.":PRINT
:rem 82
280 PRINT"2. SCEGLIERE UNA SERIE DI AZIONI TRA":rem 120
290 PRINT"{ 3 SPAZI}VARIE ALTERNATIVE.":PRINT:rem 23
300 PRINT"3. DECIDERE 'SI' O 'NO'.":PRINT:rem 147
320 PRINT"QUALE (1-2-3)?" :rem 222
330 GETR$:IFR$=""THEN330 :rem 113
340 T=VAL(R$):IFT<1ORT>3THEN330 :rem 135
350 PRINTT:GOSUB2000 :rem 245
400 FORJ=1TO9:GETR$:NEXT:ONTGOTO410,440,470 :rem 156
410 PRINT"SU CHE TIPO DI QUESTIONE DEVI DECIDERE"
:rem 65
420 INPUTT$:GOTO500 :rem 166
440 T$="SERIE DI AZIONI":GOTO500 :rem 104
470 T$="'SI' O 'NO'" :rem 184
480 NI=2:L$(1)="DECIDERE DI SI":L$(2)="DECIDERE DI NO"
:rem 186
490 GOTO750 :rem 114
500 GOSUB2000:NI=0 :rem 21
510 PRINT"MI SERVE UNA LISTA DI OGNI" :rem 29
520 PRINTT$;" IN ESAME.":PRINT :rem 18
530 PRINT"BATTI UNA RIGA ALLA VOLTA" :rem 4
540 PRINT"RISSPONDENDO A OGNI PUNTO DI DOMANDA.":PRINT
:rem 41
550 PRINT"NON IMPORTA L'ORDINE IN CUI BATTI." :rem 64
570 PRINT:PRINT"BATTI LA PAROLA ";E$;" PER INDICARE"
:rem 189
580 PRINT"CHE LA LISTA E' TERMINATA.":PRINT :rem 78
590 NI=NI+1:INPUTL$(NI) :rem 135
600 IFL$(NI)<>E$THEN590 :rem 45
610 NI=NI-1 :rem 96
620 IFNI>=2THEN650 :rem 53
630 PRINT:PRINT"DEVI FARE ALMENO 2 SCELTE.":PRINT:PRINT"
PROVA ANCORA.":GOSUB2100:GOTO500 :rem 20
650 GOSUB2000:PRINT"BENE, QUESTA E' LA LISTA CHE MI HAI D
ATO.":PRINT :rem 135
660 FORJ=1TONI:PRINT"{ 3 SPAZI}";J;"{SIN}) ";L$(J):NEXT:P
RINT :rem 208
670 PRINT"{GIU'}E' CORRETTA?" :rem 14
680 GETR$:IFNOT(R$="S"ORR$="N")THEN680 :rem 252
690 IFR$="S"THENPRINT"SI":GOTO750 :rem 35
700 IFR$="N"THENPRINT"NO" :rem 8
710 IFR$="N"THENPRINT:PRINT"RIBATTI LA LISTA":GOSUB2100:G
OTO500 :rem 81
730 GOTO680 :rem 113
750 GOSUB2000 :rem 222
760 PRINT"ORA PENSA AI DIFFERENTI FATTORI" :rem 202
770 IFT<3THENPRINT"CHE RITIENI IMPORTANTI PER SCEGLIERE"
:rem 85
780 IFT<3THENPRINT"IL/LA MIGLIORE ";T$;".":rem 250

```

```

790 IFT=3THENPRINT"CHE RITIENI IMPORTANTI PER DECIDERE SI
    O NO." :rem 176
800 PRINT"BATTI LA PAROLA ";E$;" PER INDICARE CHE":PRINT"
    LA LISTA E' TERMINATA." :rem 22
830 NF=0 :rem 156
840 NF=NF+1:INPUTF$(NF) :rem 118
850 IFF$(NF)<>E$THEN840 :rem 41
860 NF=NF-1 :rem 97
870 IFNF<1THENPRINT"DEVI FARE ALMENO 1 SCELTA! RIPROVA.":
    GOSUB2100:GOTO750 :rem 28
890 GOSUB2000:PRINT"ECCE LA LISTA CHE MI HAI DATO:":PRINT
    :rem 243
900 FORJ=1TONF:PRINT"{ 3 SPAZI}";J;"{SIN}";F$(J):NEXT
    :rem 253
910 PRINT:PRINT"ORA DEVI INDICARMI IL FATTORE PIU' IM- "
    :rem 206
920 PRINT"PORTANTE. BATTI 0 SE VUOI CORREGGERE LA
    { 2 SPAZI}LISTA.":PRINT :rem 234
940 INPUTA:A=INT(A):IFA=0THEN750 :rem 229
960 IFA>NFTHEN890 :rem 17
1000 GOSUB2000:IFNF=1THEN1200 :rem 192
1001 PRINT"ORA DOBBIAMO CONFRONTARE GLI ALTRI FAT- TORI."
    :rem 63
1010 PRINTF$(A);" HA ASSEGNATO UN VALORE DI 10.":rem 128
1020 PRINT"CHE VALORE HANNO GLI ALTRI FATTORI?" :rem 209
1080 FORJ=1TONF:IFJ=ATHEN1110 :rem 34
1090 PRINT:PRINTF$(J);:INPUTV(J) :rem 25
1100 IFA<0ORA>10THENPRINT"IL MASSIMO E' 10, IL MINIMO E'
    0. RIFAI.":GOTO1090 :rem 36
1110 NEXT :rem 2
1200 V(A)=10:Q=0:FORJ=1TONF:Q=Q+V(J):NEXT:FORJ=1TONF
    :rem 15
1210 V(J)=V(J)/Q:NEXT:GOSUB2000 :rem 88
1220 PRINT"ORA CONFRONTIAMO LE VARIE POSSIBILITA'."
    :rem 39
1310 PRINTL$(1)" HA ASSEGNATO UN VALORE DI 10." :rem 62
1390 FORJ=1TONF:PRINT"{ 4 SPAZI}-----"
    :rem 33
1410 PRINT"ESAMINIAMO ";F$(J);".":PRINTL$(1);" HA ASSEGNA
    TO UN VALORE DI 10." :rem 213
1440 FORK=2TONI :rem 164
1450 PRINTL$(K);:INPUTC(K,J):PRINT :rem 132
1470 NEXT:C(1,J)=10:NEXT :rem 151
1500 FORJ=1TONF:Q=0:FORK=1TONI :rem 168
1510 Q=Q+C(K,J):NEXT:FORK=1TONI :rem 179
1520 C(K,J)=C(K,J)/Q:NEXT:NEXT :rem 33
1530 FORK=1TONI:D(K)=0:FORJ=1TONF :rem 58
1540 D(K)=D(K)+C(K,J)*V(J):NEXT:NEXT :rem 84
1550 MX=0:FORK=1TONI :rem 241
1560 IFD(K)>MXTHENMX=D(K) :rem 15

```

```

1570 NEXT:FORK=1TONI:D(K)=D(K)*100/MX:NEXT      :rem 95
1600 FORK=1TONI:Z(K)=K:NEXT:NM=NI-1             :rem 217
1610 FORK=1TONI:FORJ=1TONM:N1=Z(J):N2=Z(J+1):IFD(N1)>D(N2)
)THEN1630                                         :rem 21
1620 Z(J+1)=N1:Z(J)=N2                           :rem 194
1630 NEXT:NEXT:J1=Z(1):J2=Z(2):DF=D(J1)-D(J2):GOSUB2000
                                                :rem 235
1640 PRINT"DOPO LE VARIE ELABORAZIONI"          :rem 213
1700 PRINTL$(J1);" RISULTA LA MIGLIORE ":PRINT"SCELTA."
                                                :rem 14
1800 PRINT:PRINT"ECCO, NELL'ORDINE, LA LISTA FINALE.":PRINT
NT                                                :rem 20
1810 PRINT"A ";L$(J1);" E' STATO DATO":PRINT"UN VALORE DI
100."                                             :rem 52
1840 PRINT"{ 4 SPAZI }-----":PRINT
RINT                                             :rem 50
1850 FORJ=1TONI:Q=Z(J):PRINTD(Q);TAB(16);L$(Q):NEXT:END
                                                :rem 178
1860 END                                          :rem 166
2000 PRINT"{RED}{CLR}{ 16 SPAZI}{RVS}DECIDERE{ 2 GIU' }
{OFF}{BLK}":RETURN                              :rem 231
2100 FORJ=1TO2500:NEXT:RETURN                   :rem 95

```



L'accoppiata

In 'L'accoppiata', è necessario, come dice il nome, formare delle coppie con un certo numero di carte coperte.

In questa versione per computer è possibile allenarsi giocando da soli o sfidare un altro giocatore in una lotta fino all'ultima coppia.

Facendo partire il gioco, dopo alcuni secondi in cui il programma inizializza alcune variabili, viene chiesto all'utente se desidera giocare da solo o contro un avversario.

Dopo aver risposto a tale domanda appaiono, una per una, sullo schermo le 36 carte che compon-

gono di conseguenza 18 coppie. Esse vengono disposte casualmente sullo schermo e in ogni partita sono in posizioni diverse.

Dopo che le carte sono state poste sullo schermo, un giocatore dovrà scegliere una coppia di carte, indicandole con i numeri che recano sul retro (se si è selezionato il solitario sarà sempre lo stesso giocatore a scegliere, se invece si è selezionata la partita a due, sceglierà a turno uno o l'altro giocatore). Dopo aver digitato i due numeri (o digitandoli separati da una virgola o premendo RETURN dopo aver digitato ciascun numero), il com-

puter volterà le due carte corrispondenti.

Se esse sono uguali, il giocatore che le ha scelte si aggiudica un punto, se sono diverse, il computer le lascia vedere per alcuni attimi e poi le volta. Se si indica una carta già scoperta, la scelta viene annullata e viene di nuovo formulata la richiesta dei due numeri.

Al termine della partita, cioè quando tutte le coppie sono state scoperte, colui il quale ha scoperto

più coppie vince.

In questo gioco è necessario, per scoprire più coppie possibile, osservare e cercare di ricordare le posizioni delle carte, perché se si vede una carta e ci si ricorda dove era la carta gemella, si può formare una coppia, mentre digitando numeri a caso, è molto difficile trovarne una.

Ed ora buon divertimento e buona accoppiata!!

```
10 POKE53280,3:POKE53281,1:PRINT" {BLU}" :rem 228
20 DIMCC(18),DC%(36),NC(72),AC(3,3),KK(18),DK%(36),SC(2) :rem 115
30 BS=1190:BC=55462 :rem 2
32 SO$="{HOME}" :rem 191
40 DATA129,130,131,132,133,134,135,136,137,138,139,140,141,142,143,144,145,146 :rem 35
50 FORI=1TO18:READIN:CC(I)=IN:NEXT :rem 163
65 DATA0,2,5,6,0,6,4,3,3,0,4,5,6,5,3,2,2,4 :rem 13
70 FORI=1TO18 :rem 17
72 READIN:KK(I)=IN :rem 90
75 NEXT :rem 171
90 FORI=0TO8:NC(I*2+1)=49+I:NC(I*2+2)=32:NEXTI :rem 158
92 FORI=9TO18:NC(I*2+1)=49:NC(I*2+2)=39+I:NEXTI :rem 225
94 FORI=19TO28:NC(I*2+1)=50:NC(I*2+2)=29+I:NEXTI :rem 12
96 FORI=29TO35:NC(I*2+1)=51:NC(I*2+2)=19+I:NEXTI :rem 13
110 FORI=1TO36 :rem 60
120 DC%(I)=0:NEXTI :rem 8
130 REM :rem 120
140 FORI=1TO18:FORJ=1TO2 :rem 237
160 P=INT(36*RND(0)+1) :rem 180
170 IFDC%(P)=0THENDC%(P)=CC(I):DK%(P)=KK(I):GOTO200 :rem 187
180 P=P+7:IFP>36THENP=P-36 :rem 58
190 GOTO170 :rem 107
200 NEXT:NEXT :rem 74
210 SC(1)=0:SC(2)=0:TR=0:PRINT" {CLR}" :rem 195
215 INPUT"GIOCATORI (1/2)";NP:IFNP<1ORNP>2THEN215 :rem 19
216 PRINT" {CLR}":ME=1 :rem 55
220 FORX=1TO36:GOSUB5300:GOSUB5100:GOSUB5200:NEXTX :rem 161
230 PRINTSO$; :rem 35
235 ME=NP+1-ME :rem 34
```



```

240 PRINT"{GIU'}{BLU}GIOCATORE";ME;"{SIN} - PRENDI 2 CART
E" :rem 173
250 INPUTX1,X2:PRINT"{SU}{ 7 SPAZI}" :rem 2
260 REM :rem 124
270 IFX1>36ORX2>36ORX1<1ORX2<1THENE$="NON BARARE !!":GOSU
B4000:GOTO230 :rem 210
280 IFDC%(X1)=0ORDC%(X2)=0ORX1=X2THENE$="GIA' SCOPERTE !"
:GOSUB4000:GOTO230 :rem 191
300 TR=TR+1:GOSUB4500 :rem 251
305 PRINTSO$;"{RED}PUNTI 1={PUR}";SC(1);"{RED}{ 3 SPAZI}
2={PUR}";SC(2) :rem 109
310 IFSC(1)+SC(2)<18THEN230 :rem 232
320 PRINT"{HOME}{GRN}{GIU'}FINITO IN";TR;"{SIN} PROVE !!"
:rem 211
330 FORW=0TO15:POKE53280,W:FORJ=1TO500:NEXTJ:NEXTW
:rem 220
340 POKE53280,3:POKE53281,1:RUN :rem 31
4000 PRINTSO$; :rem 82
4010 PRINT"{GIU'}";E$;"{ 18 SPAZI}" :rem 202
4020 FORT=1TO2000:NEXT :rem 77
4030 PRINTSO$; :rem 85
4040 PRINT"{ 2 GIU'}" :rem 187
4050 RETURN :rem 169
4500 REM :rem 173
4520 X=X1:GOSUB5000:GOSUB5100:GOSUB5200 :rem 105
4540 X=X2:GOSUB5000:GOSUB5100:GOSUB5200 :rem 108
4550 IFDC%(X1)=DC%(X2)THENSC(ME)=SC(ME)+1:DC%(X1)=0:DC%(X
2)=0:GOTO4590 :rem 1
4560 FORT=1TO1000:NEXTT :rem 169
4570 X=X1:GOSUB5100:GOSUB5300:GOSUB5200 :rem 113
4580 X=X2:GOSUB5100:GOSUB5300:GOSUB5200 :rem 115
4590 RETURN :rem 178
5000 REM :rem 169
5010 FORI=1TO3 :rem 58
5020 FORJ=1TO3 :rem 60
5030 BC(I,J)=DC%(X) :rem 239
5040 NEXTJ:NEXTI :rem 20
5050 F=255:RETURN :rem 3
5100 REM :rem 170
5110 RW=INT((X-1)/6) :rem 85
5120 CL=X-6*RW :rem 34
5130 YS=BS+120*RW+3*CL :rem 239
5135 YC=BC+120*RW+3*CL :rem 212
5140 RETURN :rem 170
5200 REM :rem 171
5210 FORI=1TO3 :rem 60
5220 FORJ=1TO3 :rem 62
5230 POKE(YS+40*(I-1)+J),BC(I,J) :rem 221
5235 POKE(YC+40*(I-1)+J),DK%(X)ANDF :rem 179
5240 NEXTJ:NEXTI :rem 22
5250 RETURN :rem 172

```

5300	REM	:rem	172
5310	FORI=1TO3	:rem	61
5320	FORJ=1TO3	:rem	63
5330	BC(I,J)=32	:rem	2
5340	NEXTJ:NEXTI	:rem	23
5345	BC(2,2)=NC(2*(X-1)+1):BC(2,3)=NC(2*(X-1)+2)	:rem	149
5350	F=0:RETURN	:rem	154



STM

STM è una abbreviazione inglese che indica la memoria breve, cioè la capacità di ricordare le cose avvenute poco tempo prima. Questo semplice programma vi dà la possibilità di migliorare la vostra memoria breve in maniera divertente. All'inizio vi viene chiesto se volete cimentarvi con le lettere dell'alfabeto, con le cifre da 0 a 9, oppure solamente con i due simboli 0 - 1. Dopo che avete fatto la vostra scelta, vi viene presentata per un tempo molto breve una se-

quenza di numeri o di lettere, che dovete imprimervi bene nella mente. Dovrete poi digitare i simboli nello stesso ordine in cui vi sono stati presentati. Se la risposta sarà esatta, vi verrà mostrata una nuova sequenza più lunga della precedente; si comincia infatti con un solo carattere, e ogni volta la sequenza si allunga di un simbolo. Potrete anche garaggiare con i vostri amici: vincerà chi avrà ricordato la sequenza più lunga impiegando il minor tempo possibile.

10	REM **STM**	:rem	225
15	POKE53280,8:POKE53281,5:PRINT"{CLR}{ 14 DES}{RVS}{BLK}		
	* S T M * "	:rem	230
20	PRINT"{ 2 GIU' }VUOI GIOCARE CON..."	:rem	252
30	PRINT"{GIU'}1: LETTERE ?"	:rem	4
40	PRINT"{GIU'}2: NUMERI ?"	:rem	193
50	PRINT"{GIU'}3: SOLO CON 0-1 ?"	:rem	158
60	PRINT"{GIU'}4: VUOI TERMINARE ?"	:rem	223
70	GETA\$:rem	172
80	ONVAL(A\$)GOTO100,110,120,620	:rem	169
90	GOTO70	:rem	9
100	A=26:B=65:GOTO200	:rem	160
110	A=10:B=48:GOTO200	:rem	155
120	A=2:B=48	:rem	104

200	INPUT" {GIU' }SEI PRONTO <RETURN> ??";K\$:rem	157
209	IFK\$=" "THEN209	:rem	228
210	C=1:TT=300	:rem	246
220	T=TI:TA=T	:rem	34
230	A\$="":PRINT	:rem	66
240	FORD=1 TOC	:rem	21
250	B\$=CHR\$ (B+RND (0) *A)	:rem	201
260	PRINTB\$:rem	139
270	A\$=A\$+B\$:rem	49
280	FORE=1 TOTT: NEXT	:rem	248
290	PRINT" {SIN} ";	:rem	68
300	NEXT	:rem	210
310	PRINT	:rem	33
350	FORW=1 TO250: NEXT: PRINT" {CLR} "	:rem	149
400	REM**RISPOSTA**	:rem	149
410	POKE198,0: INPUT"RISPOSTA";AN\$:rem	51
420	IFAN\$=A\$THEN600	:rem	63
430	PRINT" {GIU' } {RVS} SBAGLIATO! {GIU' }"	:rem	83
440	PRINT" PUNTI: ";C	:rem	177
450	PRINT" LA RISPOSTA ERA: ";A\$:rem	30
460	PRINT" TEMPO: ";STR\$ (INT ((TA-T) /.6) /100); " SECONDI "	:rem	77
500	PRINT" {GIU' } GIOCHI ANCORA (S/N) ?"	:rem	62
510	GETA\$:rem	219
520	IF A\$="S" THEN200	:rem	32
530	IF A\$<>"N" THEN510	:rem	93
540	GOTO10	:rem	51
600	C=C+1: TA=TI	:rem	94
610	GOTO230	:rem	101
620	END	:rem	111


PRECISAZIONE

Nell'indice del n. 14 il programma "PRESET 22" per VIC 20 portava per errore la seguente dicitura: "di D. Cooper - trad. e adatt. di V. Silvestroni". Precisiamo che il sig. Vincenzo Silvestroni è l'autore di tale programma e non il traduttore. Nella testata di "PRESET 22" non veniva inoltre indicato che il programma necessita dell'espansione da 3K.

La Redazione

FINALMENTE!

La Softrivista che ti gasa!



**QUALCOSA DI SUPER, DI INEDITO,
DI IRRESISTIBILE**

IL VERO GIOCO COMINCIA ADESSO

IN EDICOLA JACKSON SOFT SERIE ORO

I giochi esclusivi per
Commodore 64 e Spectrum 48 K
importati dall'Inghilterra, mai
presentati in Italia.
Una sfida Jackson al già visto, al
già fatto, al... già registrato.



La prima
puntata del
fantastico,
inedito
PYJAMARAMA

Corri in edicola, il vero gioco comincia solo adesso
e se sei davvero bravo partecipa alla "sfida al campione",
utilizzando il tagliando che troverai sull'ultima pagina
di copertina di ogni numero.



GRUPPO
EDITORIALE
JACKSON